

Zgjidhja e ekuacioneve algjebrike dhe joalgjebrike në Maxima

Qëllimet dhe objektivat

- Komanda dhe teknika për zgjidhje ekuacionesh algjebrike dhe joalgjebrike
- Shembuj zbatimesh

Përmbajtja

- 1 Funksioni `solve`
- 2 Funksionet `allroots`, `realroots`, `find_root`
- 3 Funksioni `linsolve`
- 4 Funksioni `mnewton`

Funksioni solve

- Funksioni solve mund të zbatohet për zgjidhjen e një ose më tepër ekuacionesh.

- Menyja *Equations > Solve...*

- Zgjidhjet e ekuacioneve të rendit jo më të madh se 4:

```
solve(a*x^2+b*x+c=0, [x]);
```

```
solve(a*x^3+b*x^2+c*x+d=0, [x]);
```

- Manipulimi me zgjidhjet:

```
sol: solve(3*x^3+5*x^2+6*x+2=0, [x]);
```

```
sol[1];
```

```
float(rhs(sol[1]));
```

```
rectform(%);
```

```
makelist(rectform(float(rhs(sol[k]))), k, 1, 3);
```

Shembull: Ekuacioni i zbërthimit radioaktiv

Shembull

Të zgjidhet sipas kohës t ekuacioni i zbërthimit radioaktiv

$$q = q_0 e^{-\frac{t}{\tau}}.$$

Sa është koha e gjysmëzbërthimit të një substance radioaktive me kohë jete mesatare $\tau = 100$ vjet

Zgjidhje

```
deq : q=q0*exp(-t/tau);  
sol : solve(deq,t);  
subst([q0=1,q=0.5,tau=100],sol);
```

Shembull: Lëvizja dydimensionale me nxitim konstant

Shembull (Lëvizja dydimensionale me nxitim konstant)

Shqyrtojmë lëvizjen dydimensionale në rrafshin x - y me nxitim konstant në drejtimin y të një grimce. Ekuacionet e grimcës në drejtimet x dhe y në një kohë t jepen me

```
eqx : x=x0+v0*cos(theta)*t;  
eqy : y=y0+v0*sin(theta)*t+a*t^2/2;  
sol : solve([eqx,eqy], [v0, theta]);  
sol : solve([eqx,eqy], [v0, sin(theta)]);
```

Shembull (Vazhdim)

Shembull (Lëvizja dydimensionale me nxitim konstant...)

Gjejmë shpejtësinë fillestare v_0 dhe këndin θ ashtu që $x_0 = y_0 = 0$,
 $x = 8$ m, $y = 4$ m, $a = -9.81 \frac{\text{m}}{\text{s}^2}$, $t = 2$ s

```
soleval : subst([x0=0, x=8, y0=0, y=4, a=-9.8, t=2],  
               sol);  
th:atan(2.95);  
float(%/%pi);  
subst(theta=th, soleval[1][1]);  
v:rhs(%);  
eqsolx : subst([x0=0, v0=v, theta=th], eqx);  
eqsoly : subst([y0=0, a=-9.8, v0=v, theta=th], eqy);  
wxplot2d([parametric, rhs(eqsolx), rhs(eqsoly)],  
         [t,0,4]);
```

Funksionet allroots, realroots

- Menyja *Equations > Roots of Polynomial (...)*
- Funksioni allroots mund të zbatohet për kalkulimin e të gjitha rrënjëve (reale dhe komplekse) të një ekuacioni polinomial.

```
allroots(3*x^3+5*x^2+6*x+2=0);
```

```
allroots(6*x^5-x^4+3*x^3+x-x+125=0);
```

- Funksioni realroots kthen të gjitha rrënjët reale të një ekuacioni polinomial.

```
realroots(3*x^3+5*x^2+6*x+2=0);
```

```
realroots(6*x^5-x^4+3*x^3+x-x+125=0);
```

- Funksionet allroots dhe realroots kërkojnë polinom specifik për të kthyer zgjidhjet:

```
allroots(a*x^2+b*x+c=0);
```


Funksioni find_root

- Menyja *Equations > Find Root...*
- Funksioni find_root përdoret për zgjidhje numerike të një ekuacioni $f(x) = 0$ në një interval $[a, b]$:
`find_root(cos(x)=x, x, 0, %pi/2);`

Shembull: Trajektorja e një projektili

Shembull (Trajektorja e një projektili)

Një projekt il i lansuar në fushën gravitacionale të Tokës me pozitë x përgjat drejtimit horizontal dhe pozitë y përgjat drejtimit vertikal përshkruan trajektoren e dhënë me ekuacionin vijues

```
eliminate([eqx, eqy],[t]);  
solve(%[1], y);  
eqyt : expand(%[1]);
```

Shembull (Vazhdim)

Shembull (Trajektorja e një projektili ...)

Supozojmë se janë dhënë vlerat vijuese të të dhënave: $x_0 = 5$ m, $y_0 = 20$ m, $x = 20$ m, $y = 12$ m, $v_0 = 10 \frac{\text{m}}{\text{s}}$, $a = -9.81 \frac{\text{m}}{\text{s}^2}$. Gjejmë këndin θ :

```
eqt1 : subst([x0=5, y0=20, x=20, y=12, v0=10, a=-9.81],
             eqyt);
f(theta):=rhs(eqt1)-lhs(eqt1);
f(theta);
wxplot2d(f(theta), [theta,-2*%pi,2*%pi], [y,-100,100],
         [xlabel,"theta"], [ylabel,"f(theta)"]);
wxplot2d(f(theta), [theta,0,1], [y,-10,10],
         [xlabel,"theta"], [ylabel,"f(theta)"]);
theta_1:find_root(eqt1, theta, 0, 0.5);
theta_2:find_root(eqt1, theta, 0.5, 1);
```

Funksioni linsolve

- Menyja *Equations > Solve Linear System...*
- Funksioni linsolve përdoret për të zgjidhur një sistem ekuacionesh lineare:

```
linsolve([x1+2*x2+3*x3+4*x4=25, -x2+x3-x4=5,
          2*x1+3*x3=19, x1+2*x2+2*x3=20],
          [x1,x2,x3,x4]);
a:matrix([1,2,3,4], [0,-1,1,-1], [2,0,3,0],
          [1,2,2,0]);
b:matrix([25],[5],[19],[20]);
a^^-1;
a^^-1 . b;
float(%);
```

Shembull: Sistem i pacaktuar

Shembull (Sistem i pacaktuar)

Zgjidhim sistemin vijues të dy ekuacioneve lineare me tri të panjohura

```
linsolve([x+y+z=6, x+2*y+z=8],[x,y,z]);  
wxplot3d([4-r,2,r], [r,-2,2], [s,-2,2]);
```

Shembull: Sistem i pamundur

Shembull (Sistem i pamundur)

Zgjidhim sistemin vijues të tri ekuacioneve lineare me dy të panjohura

```
linsolve([x+y=5, x+2*y=8, 3*x+y=9],[x,y]);  
linsolve([x+y=5, x+2*y=8, 3*x+y=10],[x,y]);
```

Shembull: Sistem parametrik

Shembull (Sistem parametrik)

Zgjidhim sistemin vijues parametrik:

```
tmp : linsolve([x-2*y+3*z=1, 2*x+t*y+6*z=6,  
               -x+3*y+(t-3)=0],  
               [x,y,z]);  
wxplot3d([rhs(tmp[1]), rhs(tmp[2]), rhs(tmp[3])],  
          [t,-3.7,-2], [s,-1,1]);  
wxplot3d([rhs(tmp[1]), rhs(tmp[2]), rhs(tmp[3])],  
          [t,-6,-4.3], [s,-1,1]);
```

Funksioni mnewton

- Funksioni mnewton nga pakaja mnewton lejon zgjidhjen numerike të sistemeve të ekuacioneve jolineare:

```
eq1 : x*log(y)+y*log(x)=4;
```

```
eq2 : x^2*exp(-y)+x*y=6;
```

```
wxplot3d(lhs(eq1), [x,-2,2], [y,-2,2]);
```

```
wxplot3d(lhs(eq2), [x,-2,2], [y,-2,2]);
```

```
load(mnewton);
```

```
mnewton([eq1,eq2],[x,y],[0.5,0.5]);
```

```
mnewton([eq1,eq2],[x,y],[1.0,0.5]);
```

```
mnewton([eq1,eq2],[x,y],[0.5,1.0]);
```

```
mnewton([eq1,eq2],[x,y],[1.0,1.0]);
```

```
mnewton([eq1,eq2],[x,y],[1.5,1.0]);
```


Përfundim

- Zgjidhja analitike e ekuacioneve me anë të funksionit solve
- Kompjutimi numerik i rrënjëve të një polinomi me anë të funksioneve allroots dhe realroots
- Zgjidhja e një sistemi ekuacionesh lineare me anë të funksionit linsolve
- Zgjidhja e një sistemi ekuacionesh jolineare me anë të funksionit mnewton nga pakoja homonime